Rules for normalizing to known sine integrands

1. $\int u\,(c\,\text{Trig}[a+b\,x])^m\,(d\,\text{Trig}[a+b\,x])^n\,dx$ when KnownSineIntegrandQ[u, x]

   1. $\int u\,(c\,\text{Tan}[a+b\,x])^m\,(d\,\text{Trig}[a+b\,x])^n\,dx$ when KnownSineIntegrandQ[u, x]

      1: $\int u\,(c\,\text{Tan}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^n\,dx$ when KnownSineIntegrandQ[u, x] $\land\ m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,\text{Tan}[a+b\,x])^m\,(d\,\text{Cos}[a+b\,x])^m}{(d\,\text{Sin}[a+b\,x])^m} == 0$

Rule: If KnownSineIntegrandQ[u, x] $\land\ m \notin \mathbb{Z}$, then

$$\int u\,(c\,\text{Tan}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\text{Tan}[a+b\,x])^m\,(d\,\text{Cos}[a+b\,x])^m}{(d\,\text{Sin}[a+b\,x])^m}\int \frac{u\,(d\,\text{Sin}[a+b\,x])^{m+n}}{(d\,\text{Cos}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*tan[a_.+b_.*x_])^m_.*(d_.*sin[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Tan[a+b*x])^m*(d*Cos[a+b*x])^m/(d*Sin[a+b*x])^m*Int[ActivateTrig[u]*(d*Sin[a+b*x])^(m+n)/(d*Cos[a+b*x])^m,x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSineIntegrandQ[u,x] && Not[IntegerQ[m]]
```

2: $\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Cos}[a+b\,x])^n\,dx$ when KnownSineIntegrandQ[u, x] $\wedge$ $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Cos}[a+b\,x])^m}{(d\,\mathrm{Sin}[a+b\,x])^m} == 0$

Rule: If KnownSineIntegrandQ[u, x] $\wedge$ $m \notin \mathbb{Z}$, then

$$\int u\,(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Cos}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\mathrm{Tan}[a+b\,x])^m\,(d\,\mathrm{Cos}[a+b\,x])^m}{(d\,\mathrm{Sin}[a+b\,x])^m}\int \frac{u\,(d\,\mathrm{Sin}[a+b\,x])^m}{(d\,\mathrm{Cos}[a+b\,x])^{m-n}}\,dx$$

Program code:

```
Int[u_*(c_.*tan[a_.+b_.*x_])^m_.*(d_.*cos[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Tan[a+b*x])^m*(d*Cos[a+b*x])^m/(d*Sin[a+b*x])^m*Int[ActivateTrig[u]*(d*Sin[a+b*x])^m/(d*Cos[a+b*x])^(m-n),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSineIntegrandQ[u,x] && Not[IntegerQ[m]]
```

2. $\int u\,(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Trig}[a+b\,x])^n\,dx$ when KnownSineIntegrandQ[u, x]

   1: $\int u\,(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^n\,dx$ when KnownSineIntegrandQ[u, x] $\wedge\ m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^m}{(d\,\text{Cos}[a+b\,x])^m} == 0$

Rule: If KnownSineIntegrandQ[u, x] $\wedge\ m \notin \mathbb{Z}$, then

$$\int u\,(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^m}{(d\,\text{Cos}[a+b\,x])^m}\int \frac{u\,(d\,\text{Cos}[a+b\,x])^m}{(d\,\text{Sin}[a+b\,x])^{m-n}}\,dx$$

Program code:

```
Int[u_*(c_.*cot[a_.+b_.*x_])^m_.*(d_.*sin[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Cot[a+b*x])^m*(d*Sin[a+b*x])^m/(d*Cos[a+b*x])^m*Int[ActivateTrig[u]*(d*Cos[a+b*x])^m/(d*Sin[a+b*x])^(m-n),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSineIntegrandQ[u,x] && Not[IntegerQ[m]]
```

2: $\int u\,(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Cos}[a+b\,x])^n\,dx$ when KnownSineIntegrandQ[u, x] $\land$ $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^m}{(d\,\text{Cos}[a+b\,x])^m} == 0$

Rule: If KnownSineIntegrandQ[u, x] $\land$ $m \notin \mathbb{Z}$, then

$$\int u\,(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Cos}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\text{Cot}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^m}{(d\,\text{Cos}[a+b\,x])^m}\int \frac{u\,(d\,\text{Cos}[a+b\,x])^{m+n}}{(d\,\text{Sin}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*cot[a_.+b_.*x_])^m_.*(d_.*cos[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Cot[a+b*x])^m*(d*Sin[a+b*x])^m/(d*Cos[a+b*x])^m*Int[ActivateTrig[u]*(d*Cos[a+b*x])^(m+n)/(d*Sin[a+b*x])^m,x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSineIntegrandQ[u,x] && Not[IntegerQ[m]]
```

3: $\int u\,(c\,\text{Sec}[a+b\,x])^m\,(d\,\text{Cos}[a+b\,x])^n\,dx$ when KnownSineIntegrandQ[u, x]

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((c\,\text{Sec}[a+b\,x])^m\,(d\,\text{Cos}[a+b\,x])^m\right)=0$

Rule: If KnownSineIntegrandQ[u, x], then

$$\int u\,(c\,\text{Sec}[a+b\,x])^m\,(d\,\text{Cos}[a+b\,x])^n\,dx\;\longrightarrow\;(c\,\text{Sec}[a+b\,x])^m\,(d\,\text{Cos}[a+b\,x])^m\int u\,(d\,\text{Cos}[a+b\,x])^{n-m}\,dx$$

Program code:

```
Int[u_*(c_.*sec[a_.+b_.*x_])^m_.*(d_.*cos[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Sec[a+b*x])^m*(d*Cos[a+b*x])^m*Int[ActivateTrig[u]*(d*Cos[a+b*x])^(n-m),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSineIntegrandQ[u,x]
```

4: $\int u\,(c\,\text{Csc}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^n\,dx$ when KnownSineIntegrandQ[u, x]

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((c\,\text{Csc}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^m\right)=0$

Rule: If KnownSineIntegrandQ[u, x], then

$$\int u\,(c\,\text{Csc}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^n\,dx\;\longrightarrow\;(c\,\text{Csc}[a+b\,x])^m\,(d\,\text{Sin}[a+b\,x])^m\int u\,(d\,\text{Sin}[a+b\,x])^{n-m}\,dx$$

Program code:

```
Int[u_*(c_.*sec[a_.+b_.*x_])^m_.*(d_.*cos[a_.+b_.*x_])^n_.,x_Symbol] :=
  (c*Csc[a+b*x])^m*(d*Sin[a+b*x])^m*Int[ActivateTrig[u]*(d*Sin[a+b*x])^(n-m),x] /;
FreeQ[{a,b,c,d,m,n},x] && KnownSineIntegrandQ[u,x]
```

2. $\int u\,(c\,\text{Trig}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\land\,$KnownSineIntegrandQ[u, x]

1: $\int u\,(c\,\text{Tan}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\,\land\,$KnownSineIntegrandQ[u, x]

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,\text{Tan}[a+b\,x])^m\,(c\,\text{Cos}[a+b\,x])^m}{(c\,\text{Sin}[a+b\,x])^m} == 0$

Rule: If $m\notin\mathbb{Z}\,\land\,$KnownSineIntegrandQ[u, x], then

$$\int u\,(c\,\text{Tan}[a+b\,x])^m\,dx \longrightarrow \frac{(c\,\text{Tan}[a+b\,x])^m\,(c\,\text{Cos}[a+b\,x])^m}{(c\,\text{Sin}[a+b\,x])^m}\int \frac{u\,(c\,\text{Sin}[a+b\,x])^m}{(c\,\text{Cos}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*tan[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Tan[a+b*x])^m*(c*Cos[a+b*x])^m/(c*Sin[a+b*x])^m*Int[ActivateTrig[u]*(c*Sin[a+b*x])^m/(c*Cos[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownSineIntegrandQ[u,x]
```

2: $\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,dx$ when $m \notin \mathbb{Z}\ \land$ KnownSineIntegrandQ[u, x]

- Derivation: Piecewise constant extraction

- Basis: $\partial_x \frac{(c\,\mathrm{Cot}[a+b\,x])^m\,(c\,\mathrm{Sin}[a+b\,x])^m}{(c\,\mathrm{Cos}[a+b\,x])^m} == 0$

- Rule: If $m \notin \mathbb{Z}\ \land$ KnownSineIntegrandQ[u, x], then

$$\int u\,(c\,\mathrm{Cot}[a+b\,x])^m\,dx \longrightarrow \frac{(c\,\mathrm{Cot}[a+b\,x])^m\,(c\,\mathrm{Sin}[a+b\,x])^m}{(c\,\mathrm{Cos}[a+b\,x])^m}\int \frac{u\,(c\,\mathrm{Cos}[a+b\,x])^m}{(c\,\mathrm{Sin}[a+b\,x])^m}\,dx$$

- Program code:

```
Int[u_*(c_.*cot[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Cot[a+b*x])^m*(c*Sin[a+b*x])^m/(c*Cos[a+b*x])^m*Int[ActivateTrig[u]*(c*Cos[a+b*x])^m/(c*Sin[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownSineIntegrandQ[u,x]
```

3: $\int u\,(c\,\text{Sec}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\land$ KnownSineIntegrandQ[u, x]

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((c\,\text{Sec}[a+b\,x])^m\,(c\,\text{Cos}[a+b\,x])^m\right)==0$

Rule: If $m\notin\mathbb{Z}\land$ KnownSineIntegrandQ[u, x], then

$$\int u\,(c\,\text{Sec}[a+b\,x])^m\,dx\,\longrightarrow\,(c\,\text{Sec}[a+b\,x])^m\,(c\,\text{Cos}[a+b\,x])^m\int\frac{u}{(c\,\text{Cos}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*sec[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Sec[a+b*x])^m*(c*Cos[a+b*x])^m*Int[ActivateTrig[u]/(c*Cos[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownSineIntegrandQ[u,x]
```

4: $\int u\,(c\,\text{Csc}[a+b\,x])^m\,dx$ when $m\notin\mathbb{Z}\land\text{KnownSineIntegrandQ}[u,x]$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((c\,\text{Csc}[a+b\,x])^m\,(c\,\text{Sin}[a+b\,x])^m\right)=0$

Rule: If $m\notin\mathbb{Z}\land\text{KnownSineIntegrandQ}[u,x]$, then

$$\int u\,(c\,\text{Csc}[a+b\,x])^m\,dx\;\longrightarrow\;(c\,\text{Csc}[a+b\,x])^m\,(c\,\text{Sin}[a+b\,x])^m\int\frac{u}{(c\,\text{Sin}[a+b\,x])^m}\,dx$$

Program code:

```
Int[u_*(c_.*csc[a_.+b_.*x_])^m_.,x_Symbol] :=
  (c*Csc[a+b*x])^m*(c*Sin[a+b*x])^m*Int[ActivateTrig[u]/(c*Sin[a+b*x])^m,x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[m]] && KnownSineIntegrandQ[u,x]
```

3. $\int u\,(A+B\,\mathrm{Csc}[a+b\,x])\,dx$ when KnownSineIntegrandQ[u, x]

1: $\int u\,(c\,\mathrm{Sin}[a+b\,x])^n\,(A+B\,\mathrm{Csc}[a+b\,x])\,dx$ when KnownSineIntegrandQ[u, x]

– Derivation: Algebraic normalization

– Rule: If KnownSineIntegrandQ[u, x], then

$$\int u\,(c\,\mathrm{Sin}[a+b\,x])^n\,(A+B\,\mathrm{Csc}[a+b\,x])\,dx \longrightarrow c\int u\,(c\,\mathrm{Sin}[a+b\,x])^{n-1}\,(B+A\,\mathrm{Sin}[a+b\,x])\,dx$$

– Program code:

```
Int[u_*(c_.*sin[a_.+b_.*x_])^n_.*(A_+B_.*csc[a_.+b_.*x_]),x_Symbol] :=
  c*Int[ActivateTrig[u]*(c*Sin[a+b*x])^(n-1)*(B+A*Sin[a+b*x]),x] /;
FreeQ[{a,b,c,A,B,n},x] && KnownSineIntegrandQ[u,x]
```

```
Int[u_*(c_.*cos[a_.+b_.*x_])^n_.*(A_+B_.*sec[a_.+b_.*x_]),x_Symbol] :=
  c*Int[ActivateTrig[u]*(c*Cos[a+b*x])^(n-1)*(B+A*Cos[a+b*x]),x] /;
FreeQ[{a,b,c,A,B,n},x] && KnownSineIntegrandQ[u,x]
```

2: $\int u\,(A+B\,\mathrm{Csc}[a+b\,x])\,dx$ when KnownSineIntegrandQ[u, x]

Derivation: Algebraic normalization

Rule: If KnownSineIntegrandQ[u, x], then

$$\int u\,(A+B\,\mathrm{Csc}[a+b\,x])\,dx\;\longrightarrow\;\int\frac{u\,(B+A\,\mathrm{Sin}[a+b\,x])}{\mathrm{Sin}[a+b\,x]}\,dx$$

Program code:

```
Int[u_*(A_+B_.*csc[a_.+b_.*x_]),x_Symbol] :=
  Int[ActivateTrig[u]*(B+A*Sin[a+b*x])/Sin[a+b*x],x] /;
FreeQ[{a,b,A,B},x] && KnownSineIntegrandQ[u,x]
```

```
Int[u_*(A_+B_.*sec[a_.+b_.*x_]),x_Symbol] :=
  Int[ActivateTrig[u]*(B+A*Cos[a+b*x])/Cos[a+b*x],x] /;
FreeQ[{a,b,A,B},x] && KnownSineIntegrandQ[u,x]
```

4. $\int u\,(A + B\,\text{Csc}[a + b\,x] + C\,\text{Csc}[a + b\,x]^2)\,dx$ when KnownSineIntegrandQ[u, x]

1: $\int u\,(c\,\text{Sin}[a + b\,x])^n\,(A + B\,\text{Csc}[a + b\,x] + C\,\text{Csc}[a + b\,x]^2)\,dx$ when KnownSineIntegrandQ[u, x]

- Derivation: Algebraic normalization

- Rule: If KnownSineIntegrandQ[u, x], then

$$\int u\,(c\,\text{Sin}[a + b\,x])^n\,(A + B\,\text{Csc}[a + b\,x] + C\,\text{Csc}[a + b\,x]^2)\,dx \longrightarrow c^2 \int u\,(c\,\text{Sin}[a + b\,x])^{n-2}\,(C + B\,\text{Sin}[a + b\,x] + A\,\text{Sin}[a + b\,x]^2)\,dx$$

- Program code:

```
Int[u_.*(c_.*sin[a_.+b_.*x_])^n_.*(A_.+B_.*csc[a_.+b_.*x_]+C_.*csc[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Sin[a+b*x])^(n-2)*(C+B*Sin[a+b*x]+A*Sin[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,B,C,n},x] && KnownSineIntegrandQ[u,x]
```

```
Int[u_.*(c_.*cos[a_.+b_.*x_])^n_.*(A_.+B_.*sec[a_.+b_.*x_]+C_.*sec[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Cos[a+b*x])^(n-2)*(C+B*Cos[a+b*x]+A*Cos[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,B,C,n},x] && KnownSineIntegrandQ[u,x]
```

```
Int[u_.*(c_.*sin[a_.+b_.*x_])^n_.*(A_+C_.*csc[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Sin[a+b*x])^(n-2)*(C+A*Sin[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,C,n},x] && KnownSineIntegrandQ[u,x]
```

```
Int[u_.*(c_.*cos[a_.+b_.*x_])^n_.*(A_+C_.*sec[a_.+b_.*x_]^2),x_Symbol] :=
  c^2*Int[ActivateTrig[u]*(c*Cos[a+b*x])^(n-2)*(C+A*Cos[a+b*x]^2),x] /;
FreeQ[{a,b,c,A,C,n},x] && KnownSineIntegrandQ[u,x]
```

2: $\int u\,(A + B\,\text{Csc}[a + b\,x] + C\,\text{Csc}[a + b\,x]^2)\,dx$ when KnownSineIntegrandQ[u, x]

Derivation: Algebraic normalization

Rule: If KnownSineIntegrandQ[u, x], then

$$\int u\,(A + B\,\text{Csc}[a + b\,x] + C\,\text{Csc}[a + b\,x]^2)\,dx \to \int \frac{u\,(C + B\,\text{Sin}[a + b\,x] + A\,\text{Sin}[a + b\,x]^2)}{\text{Sin}[a + b\,x]^2}\,dx$$

Program code:

```
Int[u_*(A_.+B_.*csc[a_.+b_.*x_]+C_.*csc[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+B*Sin[a+b*x]+A*Sin[a+b*x]^2)/Sin[a+b*x]^2,x] /;
FreeQ[{a,b,A,B,C},x] && KnownSineIntegrandQ[u,x]
```

```
Int[u_*(A_.+B_.*sec[a_.+b_.*x_]+C_.*sec[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+B*Cos[a+b*x]+A*Cos[a+b*x]^2)/Cos[a+b*x]^2,x] /;
FreeQ[{a,b,A,B,C},x] && KnownSineIntegrandQ[u,x]
```

```
Int[u_*(A_+C_.*csc[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+A*Sin[a+b*x]^2)/Sin[a+b*x]^2,x] /;
FreeQ[{a,b,A,C},x] && KnownSineIntegrandQ[u,x]
```

```
Int[u_*(A_+C_.*sec[a_.+b_.*x_]^2),x_Symbol] :=
  Int[ActivateTrig[u]*(C+A*Cos[a+b*x]^2)/Cos[a+b*x]^2,x] /;
FreeQ[{a,b,A,C},x] && KnownSineIntegrandQ[u,x]
```

5: $\int u\,(A + B\,\text{Sin}[a + b\,x] + C\,\text{Csc}[a + b\,x])\,dx$

Derivation: Algebraic normalization

Rule:

$$\int u \left(A + B \sin[a + b x] + C \csc[a + b x]\right) dx \longrightarrow \int \frac{u \left(C + A \sin[a + b x] + B \sin[a + b x]^2\right)}{\sin[a + b x]} dx$$

Program code:

```
Int[u_*(A_.+B_.*sin[a_.+b_.*x_]+C_.*csc[a_.+b_.*x_]),x_Symbol] :=
  Int[ActivateTrig[u]*(C+A*Sin[a+b*x]+B*Sin[a+b*x]^2)/Sin[a+b*x],x] /;
FreeQ[{a,b,A,B,C},x]
```

```
Int[u_*(A_.+B_.*cos[a_.+b_.*x_]+C_.*sec[a_.+b_.*x_]),x_Symbol] :=
  Int[ActivateTrig[u]*(C+A*Cos[a+b*x]+B*Cos[a+b*x]^2)/Cos[a+b*x],x] /;
FreeQ[{a,b,A,B,C},x]
```

6: $\int u \left(A \sin[a + b x]^n + B \sin[a + b x]^{n+1} + C \sin[a + b x]^{n+2}\right) dx$

Derivation: Algebraic normalization

Rule:

$$\int u \left(A \sin[a + b x]^n + B \sin[a + b x]^{n+1} + C \sin[a + b x]^{n+2}\right) dx \longrightarrow \int u \sin[a + b x]^n \left(A + B \sin[a + b x] + C \sin[a + b x]^2\right) dx$$

Program code:

```
Int[u_*(A_.*sin[a_.+b_.*x_]^n_.+B_.*sin[a_.+b_.*x_]^n1_+C_.*sin[a_.+b_.*x_]^n2_),x_Symbol] :=
  Int[ActivateTrig[u]*Sin[a+b*x]^n*(A+B*Sin[a+b*x]+C*Sin[a+b*x]^2),x] /;
FreeQ[{a,b,A,B,C,n},x] && EqQ[n1,n+1] && EqQ[n2,n+2]
```

```
Int[u_*(A_.*cos[a_.+b_.*x_]^n_.+B_.*cos[a_.+b_.*x_]^n1_+C_.*cos[a_.+b_.*x_]^n2_),x_Symbol] :=
  Int[ActivateTrig[u]*Cos[a+b*x]^n*(A+B*Cos[a+b*x]+C*Cos[a+b*x]^2),x] /;
FreeQ[{a,b,A,B,C,n},x] && EqQ[n1,n+1] && EqQ[n2,n+2]
```